

```

//*****
//*      MACRO: ADRS2MRG.WCM
//*      PURPOSE: Copies the address book into a merge data file.
//*****
Application (A1; "WordPerfect"; Default; "EN")

Global AllDone := False
Global DoIt := False
Global Dlg := "ADRS2MRG"
Global ABName := ""
Global AllRecords := 1

WaitMsgInit ()

DialogDefine (Dlg; 50; 50; 194; 171; Percent!; "Address Book To Merge")
DialogAddPushButton (Dlg; "OKBtn"; 136; 8; 50; 14; OKBtn! | DefaultBtn!; "OK")
DialogAddPushButton (Dlg; "CancelBtn"; 136; 26; 50; 14; CancelBtn!; "Cancel")
DialogAddGroupBox (Dlg; "GB1"; 8; 5; 120; 104; " ")
DialogAddRadioButton (Dlg; "All"; 13; 5; 108; 11; "&All records from Address Book";
AllRecords)
DialogAddText (Dlg; "T1"; 18; 18; 100; 10; 0; "Select Address Book: ")
DialogAddListBox (Dlg; "ABList"; 18; 29; 100; 72; 0; ABName)

DialogAddGroupBox (Dlg; "GB2"; 8; 114; 120; 49; " ")
DialogAddRadioButton (Dlg; "Selected"; 13; 114; 67; 11; "Selected &records"; Selected)
DialogAddPushButton (Dlg; "SelectBtn"; 18; 129; 100; 14; 0; "&Select Records...")
DialogAddText (Dlg; "SelText"; 18; 146; 100; 10; 0; "")
DialogLoad (Dlg)
WaitMsgDisplay ("Retrieving Address Book names...")

EnableFirstGroup (True)
SetStatic (0)
ABCCount := AddressBookGetCount ()
For (x; 1; x <= ABCCount; x + 1)
    Name := AddressBookGetName (x)
    RegionAddListItem (Dlg + ".ABList"; Name)
    If (x = 1)
        RegionSelectListItem (Dlg + ".ABList"; Name)
    EndIf
EndFor

WaitMsgHide ()
DialogShow (Dlg; "ABList"; CB)

Repeat
Until (AllDone)

```

```
ABName := RegionGetSelectedText (Dlg + ".ABList")
DialogDismiss (Dlg; "OKBtn")
```

```
If (DoIt)
    WaitMsgDisplay ("Placing records in Merge file...")
    If (AllRecords)
        If (ABName <> "")
            BuildFile (ABName)
        EndIf
    Else
        BuildFile ("")
    EndIf
    WaitMsgHide ()
EndIf
```

```
WaitMsgDestroy ()
DialogDestroy (Dlg)
```

```
Quit
```

```
*****
//      PROCEDURE NAME: CB
//      INPUT: None
//      OUTPUT: None
//      USES: AllDone; EnableFirstGroup
//      DESCRIPTION: Callback function for main dialog.
*****
```

```
Procedure CB ()
Switch (CB[3])
    CaseOf "":
        If (CB[5] = 274 And CB[6] = 61536)
            AllDone := True
        EndIf
    CaseOf "All":
        EnableFirstGroup (True)
    CaseOf "Selected":
        EnableFirstGroup (False)
    CaseOf "SelectBtn":
        If (AddressBookDlg (MultipleSelection!; Reselect!))
            SetStatic (?AddressBookSelection)
        EndIf
    CaseOf "OKBtn":
        AllDone := True
        DoIt := True
    CaseOf "CancelBtn":
```

```

        AllDone := True
EndSwitch
EndProc
//*****

//*****
//    PROCEDURE NAME: EnableFirstGroup
//    INPUT: Enable (Boolean)
//    OUTPUT: None
//    USES: None
//    DESCRIPTION: Enables the first group in the dialog.
//*****
Procedure EnableFirstGroup (Enable)
RegionSetCheck (Dlg + ".All"; Enable)
RegionSetCheck (Dlg + ".Selected"; Not Enable)
RegionEnableWindow (Dlg + ".SelectBtn"; Not Enable)
RegionEnableWindow (Dlg + ".SelText"; Not Enable)
RegionEnableWindow (Dlg + ".ABList"; Enable)
EndProc
//*****

//*****
//    PROCEDURE NAME: SetStatic
//    INPUT: Number
//    OUTPUT: None
//    USES: None
//    DESCRIPTION: Sets the static text in the main dialog.
//*****
Procedure SetStatic (Number)
If (Number = 1)
    Text := Number + " Record Selected"
Else
    Text := Number + " Records Selected"
EndIf
RegionSetWindowText (Dlg + ".SelText"; Text)
EndProc
//*****

//*****
//    PROCEDURE NAME: BuildFile
//    INPUT: ABName
//    OUTPUT: None
//    USES: None
//    DESCRIPTION: Sends the token to build the merge file.
//*****
Procedure BuildFile (ABName)

```

```

TempFile := ?PathDocument + "adrs2mrg.tmp"
If (ABName <> "")
    MergeAddressBookSaveAs (TempFile; ABName)
Else
    MergeAddressBookSaveAs (TempFile)
EndIf
If (?NumberOpenDocuments < 9)
    If (Not ?DocBlank)
        FileNew ()
    EndIf
    FileInsert (TempFile; No!; Insert!)
    DeleteFile (TempFile; NoPrompts!)
Else
    MessageBox (; "Conversion Complete"; "The Address Book records have been converted
to the merge data file ^0."; HasParameters! | IconInformation!; TempFile)
EndIf
EndProc
//*****

//*****
//    PROCEDURE: WaitMsgInit ()
//    PURPOSE: Initializes macro wait messages.
//*****
PROCEDURE WaitMsgInit ()
DialogDefine ("WaitMsg"; 50; 50; 150; 36; 16; "Please Wait")
DialogAddText ("WaitMsg"; "WaitText"; 8; 14; 144; 16; 1; "")
DialogLoad ("WaitMsg")
Return
ENDPROC

//*****
//    PROCEDURE: WaitMsgDisplay (Text)
//    PURPOSE: Displays a previously defined wait message dialog.
//*****
PROCEDURE WaitMsgDisplay (Text)
RegionSetWindowText ("WaitMsg" + ".WaitText"; Text)
DialogShow ("WaitMsg";; WaitDlgCallBack)
ENDPROC

PROCEDURE WaitDlgCallBack ()
If (WaitDlgCallBack[5] = 274 And WaitDlgCallBack[6] = 61536)
    Assert (CancelCondition!)
EndIf
ENDPROC

//*****

```

```
//      PROCEDURE: WaitMsgHide
//      PURPOSE: Hides a previously defined wait message dialog.
//*****
PROCEDURE WaitMsgHide ()

DialogDismiss ("WaitMsg"; "WaitText")
ENDPROC

//*****
//      PROCEDURE: WaitMsgDestroy
//      PURPOSE: Destroys a previously defined wait message dialog.
//*****
PROCEDURE WaitMsgDestroy ()
DialogDestroy ("WaitMsg")
ENDPROC
//*****
```